Cartel Leniency and Settlements: A Joint Perspective

Peter T. Dijkstra* Jacob Seifert†

Online Appendix

This Online Appendix contains proofs as well as supplementary calculations that underlie the graphical analysis in the paper.

1 Proofs

Proof of Proposition 1

Rewriting the value of collusive strategy LXX in (7) as

$$V_{LXX} = \frac{\pi_C}{1-\delta} - \frac{\alpha\left(\pi_C - \pi_N + \gamma_L F\right)}{1-\delta},$$

we can see that this exceeds the deviation profits at the collusion stage V_D, see (5), whenever $\alpha < \alpha_{LXX}$ in (8).

Now consider deviations at the leniency stage. Given that the cartel is convicted with certainty on the basis of the other colluding firms' leniency application and settlements are not available, the profits of this deviation are given by

$$V_{\check{N}XX} = \pi_N - F + \frac{\delta}{1-\delta}\pi_N = \frac{\pi_N}{1-\delta} - F. \tag{OA1}$$

Letting $V_{\dot{L}XX}$ denote the continuation value of sticking to LXX rather than not applying for leniency, we have

$$V_{\dot{L}XX} = \pi_N - \gamma_L F + \delta V_{LXX}. \tag{OA2}$$

*Corresponding author. Email: peter.dijkstra@acm.nl. Netherlands Authority for Consumers & Markets, PO Box 16326, 2500 BH The Hague, The Netherlands.

†E-mail: jacob.seifert@leicester.ac.uk. University of Leicester, School of Business, Department of Economics, Finance and Accounting, Leicester, LE1 7RH, UK.

Comparing (OA1) and (OA2), we see that $V_{\dot{L}XX} > V_{\check{N}XX}$ whenever

$$F(1-\gamma_L) + \delta\left(V_{LXX} - \frac{\pi_N}{1-\delta}\right) > 0.$$

Since $\pi_D > \pi_N$, it follows that $V_D = \pi_D + \frac{\delta}{1-\delta}\pi_N > \frac{\pi_N}{1-\delta}$. Hence, since $\gamma_L < 1$, whenever $V_{LXX} > V_D$ so that collusive strategy LXX is robust to deviations at the collusion stage, it must also be robust to deviations at the leniency stage.

The same result holds if the deviating firm can settle its case. Then, the profits of deviating at the leniency stage are given by

$$V_{\check{N}SS} = \pi_N - \gamma_S F + \frac{\delta}{1-\delta}\pi_N = \frac{\pi_N}{1-\delta} - \gamma_S F.$$

Comparing this to the continuation profits in (OA2), we see that $V_{\dot{L}XX} > V_{\check{N}SS}$ whenever

$$F(\gamma_S - \gamma_L) + \delta\left(V_{LXX} - \frac{\pi_N}{1-\delta}\right) > 0,$$

which must hold by identical arguments when LXX is robust to deviations at the collusion stage, that is when $V_{LXX} > V_D > \frac{\pi_N}{1-\delta}$, since $\gamma_S > \gamma_L$. Thus, preventing firms that deviate from collusive strategy LXX at the leniency stage from settling their case implies no loss in generality. In both cases, the binding constraint on the incentive compatibility of LXX arises at the collusion stage, not the leniency stage. □

Proof of Proposition 2

Rewriting the value of collusive strategy NNN in (9) as

$$V_{NNN} = \frac{\pi_C}{1-\delta} - \frac{\delta}{1+\delta}\frac{\alpha\, p\,(\pi_C - \pi_N + F)}{1-\delta},$$

we see that this exceeds the deviation profits attainable at the collusion stage V_D in (5) whenever $\alpha < \alpha_{NNN}$ in (11).

The second possible deviation involves a firm applying for leniency after the CA has opened an investigation, knowing it will be the first to do so. This deviation has a value

$$V_{\check{L}NN} = \pi_N - \gamma_1 F + \frac{\delta}{1-\delta}\pi_N = \frac{\pi_N}{1-\delta} - \gamma_1 F,$$

where $\gamma_1 F$ is the fine payable by the first leniency applicant. Letting $V_{\dot{N}NN}$ denote the continuation value of sticking to NNN after the investigation has been opened but the strength of the preliminary case has not yet been established (that is of *not* deviating to a leniency application), we have

$$
\begin{aligned}
V_{\dot{N}NN} &= \pi_C + \delta\big[p(\pi_N - F) + (1-p)\pi_C + \delta V_{NNN}\big] \\
&= \frac{\pi_C}{1-\delta} - \frac{1-\delta^2(1-\alpha)}{1-\delta^2} p\,\delta\,(\pi_C - \pi_N + F)\,.
\end{aligned}
$$

Sticking to NNN is more profitable if and only if $V_{\dot{N}NN} > V_{\check{L}NN}$, which is equivalent to $\alpha < \alpha_{\check{L}NN}$ in (12).

The third possible deviation involves a firm waiting until the strength of the CA's preliminary case is known and settling if the CA has a strong case. This yields a value of

$$
V_{N\check{S}N} = \pi_N - \gamma_S F + \frac{\delta}{1-\delta}\pi_N = \frac{\pi_N}{1-\delta} - \gamma_S F. \tag{OA3}
$$

Again using a dot to denote the continuation profits, sticking to the strategy of not settling when the CA has formed a strong preliminary case, so that it must complete its full prosecution, yields a value of

$$
V_{N\dot{N}N} = \pi_C + \delta\big[p_1(\pi_N - F) + (1-p_1)\pi_C + \delta V_{NNN}\big]. \tag{OA4}
$$

Sticking to NNN is more profitable if and only if $V_{N\dot{N}N} > V_{N\check{S}N}$, which is equivalent to $\alpha < \alpha_{N\check{S}N}$ in (13).

Finally, a firm could also wait until the strength of the case is known and settle if the CA has a weak case. This deviation yields a value of

$$
V_{NN\check{S}} = \pi_N - \gamma_S F + \frac{\delta}{1-\delta}\pi_N = \frac{\pi_N}{1-\delta} - \gamma_S F. \tag{OA5}
$$

The value of sticking to the strategy of not cooperating with the CA yields a value of

$$
V_{NN\dot{N}} = \pi_C + \delta\big[p_0(\pi_N - F) + (1-p_0)\pi_C + \delta V_{NNN}\big]. \tag{OA6}
$$

Since $p_0 < p_1$ implies that $V_{NN\dot{N}} > V_{N\dot{N}N}$, see (OA4) and (OA6), and since (OA3) and (OA5) imply that $V_{NN\check{S}} = V_{N\check{S}N}$, this latter deviation can be ruled out whenever $\alpha < \alpha_{N\check{S}N}$ in (13) holds. Hence, this latter deviation does not result in any additional restrictions on the incentive compatibility of NNN. $\square$

Proof of Proposition 3

Rewriting the value of collusive strategy NSN in (14) as

$$V_{NSN} = \frac{\pi_C}{1-\delta} - \frac{\alpha}{2+\delta}\frac{(\pi_C - \pi_N)(1+\delta p_0) + F(\gamma_S + \delta p_0)}{1-\delta}, \tag{OA7}$$

we see that this exceeds the deviation profits attainable at the collusion stage V_D in (5) whenever $\alpha < \alpha_{NSN}$ in (16).

The second possible deviation involves a firm applying for leniency after the CA has opened an investigation, knowing it will be the first to do so. The value of this deviation is equal to

$$V_{\check{L}SN} = \pi_N - \gamma_1 F + \frac{\delta}{1-\delta}\pi_N = \frac{\pi_N}{1-\delta} - \gamma_1 F,$$

where $\gamma_1 F$ is the fine payable by the first leniency applicant. The continuation value of sticking to NSN once the investigation has been launched is

$$\begin{aligned} V_{\dot{N}SN} = & \frac{1}{2}\left[\pi_N - \gamma_S F + \delta V_{NSN}\right] \\ & + \frac{1}{2}\Big[\pi_C + \delta\big[p_0(\pi_N - F) + (1-p_0)\pi_C + \delta V_{NSN}\big]\Big], \end{aligned}$$

where V_{NSN} is given in (OA7). Sticking to NSN is more profitable if and only if $V_{\dot{N}SN} > V_{\check{L}SN}$, which is equivalent to $\alpha < \alpha_{\check{L}SN}$ in (17).

The third possible deviation occurs when a firm waits until the strength of the CA's preliminary case is known and settles when the CA has a weak case. The value of this deviation, conditional on a weak case having been established, is

$$V_{NS\check{S}} = \pi_N - \gamma_S F + \frac{\delta}{1-\delta}\pi_N = \frac{\pi_N}{1-\delta} - \gamma_S F.$$

Sticking to NSN once the weak preliminary case is established, meanwhile, yields continuation value

$$V_{NS\dot{N}} = \pi_C + \delta\Big[p_0(\pi_N - F) + (1-p_0)\pi_C + \delta V_{NSN}\Big],$$

where V_{NSN} is given in (14). Sticking to NSN is more profitable whenever $V_{NS\dot{N}} > V_{NS\check{S}}$, which is equivalent to $\alpha < \alpha_{NS\check{S}}$ in (18).

Finally, a firm could decide unilaterally not to settle a strong case. If the CA has established a strong preliminary case, the value of not settling equals

$$V_{N\check{N}N} = \pi_N - F + \frac{\delta}{1-\delta}\pi_N = \frac{\pi_N}{1-\delta} - F.$$

If the firm would stick to settling a strong case, the continuation value equals

$$V_{N\dot{S}N} = \pi_N - \gamma_S F + \delta V_{NSN}.$$

Sticking to NSN is more profitable whenever $V_{N\dot{S}N} > V_{N\check{N}N}$, which is equivalent to

$$(1-\gamma_S)F + \delta\left(V_{NSN} - \frac{\pi_N}{1-\delta}\right) > 0.$$

The first term is non-negative because $\gamma_S \leq 1$. Hence, this condition is always satisfied if the second term is positive. This must be true whenever $\alpha < \alpha_{NSN}$, since then $V_{NSN} > V_D > \frac{\pi_N}{1-\delta}$, see (5). Hence, this latter deviation does not impose any additional restrictions on the incentive compatibility of NSN. □

Proof of Proposition 5

Algebraic comparison of (24) with (26) and (27) reveals that $W_{NC} > W_{NNN}$ and $W_{NC} > W_{NSN}$ for all feasible parameter values. Considering the comparison of (24) and (25), we can see from (8) that α_{LXX} takes its maximal value when F is small. As F tends to zero, α_{LXX} approaches

$$\frac{\pi_C - (1-\delta)\pi_D - \delta\pi_N}{\pi_C - \pi_N},$$

which lies below 1 since $\pi_D > \pi_N$. Hence, whenever LXX is an equilibrium, so that $\alpha < \alpha_{LXX} < 1$, (24) and (25) imply that $W_{NC} > W_{LXX}$. □

Proof of Proposition 7

Maximizing the size of the NC region involves shifting down the incentive compatibility constraints in Propositions 1-3. Whenever a change in a policy parameter, all other things being equal, shifts down at least one incentive compatibility constraint without shifting up any other, the size of the NC region weakly increases. From (12) and (17), both $\alpha_{\check{L}NN}$ and $\alpha_{\check{L}SN}$ are increasing in γ_1 for fixed p. The remaining incentive compatibility constraints are independent of γ_1. Hence deterrence is most likely to be achieved by setting $\gamma_1 = 0$. From (8), α_{LXX} is weakly decreasing in F (strictly so whenever $\gamma_L > 0$), while all the other incentive compatibility constraints in Propositions 2 and 3 are strictly decreasing in F. Hence the policy most likely to achieve deterrence sets F at its maximum possible level $\overline{F}$. □

Proof of Proposition 8

Since cartel deterrence is not feasible, the budget constraint must lie below the NC region for all feasible parameter values. Changes in the levels of the policy parameters therefore influence the choice of non-deterred cartels as to which collusive strategy to play in equilibrium via the profitability thresholds in (19)-(21).

The second-best policy maximizes the size of the LXX region for the following reason. Increasing the size of the LXX region implies that, if the optimal α and p previously fell within the NNN or NSN regions according to Proposition 6, this optimum point may now fall within the LXX region. This increases welfare given Lemma 2. If the optimum falls within the LXX region to begin with, increasing the size of the LXX region by shifting its left-hand boundary leftwards implies an increase in welfare because, given the downward-sloping budget constraint, we move to a higher iso-welfare curve (Figure 4 illustrates this property). Given Proposition 4, since p_{LXX}^{NNN} and p_{LXX}^{NSN} are both increasing in γ_L and p_{LXX}^{NSN} is decreasing in γ_S, see (19) and (21), the size of the LXX region is maximized by setting $\gamma_L = 0$ and $\gamma_S = 1$.

Whenever (α^*, p^*), as determined according to Proposition 6, does not fall within the LXX region when $\gamma_L = 0$ and $\gamma_S = 1$, the CA should maximize the size of the NSN region relative to the NNN region. This cannot imply a shift in the optimum point from LXX to NSN (a welfare reduction) since, by construction, the optimum point did not fall within the LXX region, even when its size was at its maximum. The size of the NSN region is maximized relative to NNN by letting $\gamma_S \to 0$. From (20), this shifts p_{NSN}^{NNN} leftwards. This increases welfare whenever the optimum point was previously in the NNN region and that point now falls within the NSN region (Lemma 2). If the optimum was already obtained within the NSN region, expanding the size of this region leaves the resulting welfare gain unchanged if NSN was implemented at an interior solution and increases welfare if NSN was implemented at a corner solution at its left-hand boundary, where $p = p_{NSN}^{NNN}$.

Since p_{LXX}^{NNN} and p_{LXX}^{NSN} are strictly decreasing in F, while p_{NSN}^{NNN} is weakly decreasing in F (and strictly so for $\gamma_S < 1$) , by identical arguments, the second-best sets $F = \overline{F}$. This maximizes the size of the LXX region relative to both NSN and NNN, and maximizes the size of the NSN region relative to NNN. □

2 Supplementary Calculations

These supplementary calculations were created in Wolfram Mathematica. The calculations file in .nb format is available on request from the authors and by download from *https://doi.org/10.25392/leicester.data.20448864*.

Incentive Compatibility and Profitability Thresholds

We first recalculate the incentive compatibility α-thresholds and profitability p-thresholds presented in the paper. These are the basis of the graphical analysis that follows.

We store the functional forms for p_0 and p_1 for future use

```
P1[d_, p_] := p + d
```

```
P0[d_, p_] := p - d
```

To ease the notational burden in these Supporting Calculations, we let lower case letters in the subscripts represent the breve/dot values in the deviation/continuation profits.

Incentive Compatibility - Collusive Strategy LXX

The profits from collusive strategy LXX can be solved as

```
Solve[V == α (πN - γL F + δ V) + (1 - α) (πC + δ V), V]
```

$$\left\{\left\{V \to \frac{F\,\alpha\,\gamma L - \pi C + \alpha\,\pi C - \alpha\,\pi N}{-1 + \delta}\right\}\right\}$$

$$\text{VLXX}[F\_, \alpha\_, \delta\_, \gamma L\_, \pi C\_, \pi N\_] := \frac{F\,\alpha\,\gamma L - \pi C + \alpha\,\pi C - \alpha\,\pi N}{-1 + \delta}$$

Deviations at the Collusion Stage

The profits from deviating at the cartel formation stage are

$$\text{VD}[\delta\_, \pi D\_, \pi N\_] := \pi D + \frac{\delta}{1 - \delta}\,\pi N$$

Hence the threshold value of α may be solved as

```
FullSimplify[Solve[VLXX[F, α, δ, γL, πC, πN] == VD[δ, πD, πN], α]]
```

$$\left\{\left\{\alpha \to \frac{\pi C + (-1 + \delta)\,\pi D - \delta\,\pi N}{F\,\gamma L + \pi C - \pi N}\right\}\right\}$$

We store this as the α_{LXX} constraint.

$$\alpha\text{LXX}[\delta\_, \gamma L\_, \pi C\_, \pi D\_, \pi N\_, F\_] := \frac{\pi C + (-1 + \delta)\,\pi D - \delta\,\pi N}{F\,\gamma L + \pi C - \pi N}$$

Incentive Compatibility - Collusive Strategy NNN

The profits from collusive strategy NNN can be solved as

```
FullSimplify[
 Solve[V == α (πC + δ (p (πN - F) + (1 - p) πC + δ V)) + (1 - α) (πC + δ (πC + δ V)), V]]
```

$$\left\{\left\{V \to -\frac{\pi C + \delta\,\pi C - p\,\alpha\,\delta\,(F + \pi C - \pi N)}{-1 + \delta^2}\right\}\right\}$$

```
VNNN[F_, p_, α_, δ_, πC_, πN_] := - (πC + δ πC - p α δ (F + πC - πN)) / (-1 + δ^2)
```

Deviations at the Collusion Stage

The deviation profits are the same as those considered for collusive strategy LXX. Hence we may solve for the relevant threshold as

```
FullSimplify[Solve[VNNN[F, p, α, δ, πC, πN] == VD[δ, πD, πN], α]]
```

$$\left\{\left\{\alpha \to \frac{(1+\delta)\,(\pi C + (-1+\delta)\,\pi D - \delta\,\pi N)}{p\,\delta\,(F + \pi C - \pi N)}\right\}\right\}$$

We store this as the α_{NNN} constraint.

```
αNNN[F_, p_, δ_, πC_, πD_, πN_] := ((1 + δ) (πC + (-1 + δ) πD - δ πN)) / (p δ (F + πC - πN))
```

Deviations at the Leniency Stage

The profits from deviating and applying for leniency as the first applicant are

```
VlNN[F_, δ_, γ1_, πN_] := πN / (1 - δ) - γ1 F
```

The continuation profits from sticking to NSN instead of applying for leniency are

```
VnNN[F_, p_, V_, δ_, πC_, πN_] := πC + δ (p (πN - F) + (1 - p) πC + δ V)
```

Hence the deviation threshold in terms of α may be calculated as

```
FullSimplify[
 Solve[VnNN[F, p, VNNN[F, p, α, δ, πC, πN], δ, πC, πN] == VlNN[F, δ, γ1, πN], α]]
```

$$\{\{\alpha \to ((1+\delta)\,(F\,(-1+\delta)\,(-\gamma 1 + p\,\delta) + (1 + p\,(-1+\delta)\,\delta)\,(\pi C - \pi N)))\,/\,(p\,\delta^3\,(F + \pi C - \pi N))\}\}$$

We store this as the α_{lNN} constraint.

```
αlNN[F_, p_, δ_, γ1_, πC_, πN_] :=
  ((1 + δ) (F (-1 + δ) (-γ1 + p δ) + (1 + p (-1 + δ) δ) (πC - πN))) / (p δ^3 (F + πC - πN))
```

Deviations at the Settlement Stage - Settling a Strong Case

The value of this deviation is

```
VNsN[F_, δ_, γS_, πN_] := πN / (1 - δ) - γS F
```

The continuation profits of sticking to NNN instead of settling the strong case are

```
VNnN[F_, p1_, V_, δ_, πC_, πN_] := πC + δ (p1 (πN - F) + (1 - p1) πC + δ V)
```

So we may solve for the α threshold as follows

```
FullSimplify[
 Solve[VNnN[F, p1, VNNN[F, p, α, δ, πC, πN], δ, πC, πN] == VNsN[F, δ, γS, πN], α]]
```

$$\{\{\alpha \to ((1+\delta)\,(F\,(-1+\delta)\,(-\gamma S + p1\,\delta) + (1 + p1\,(-1+\delta)\,\delta)\,(\pi C - \pi N)))\,/\,(p\,\delta^3\,(F + \pi C - \pi N))\}\}$$

We store this as the α_{NsN} constraint.

```
αNsN[F_, p_, p1_, δ_, γS_, πC_, πN_] :=
  ((1 + δ) (F (-1 + δ) (-γS + p1 δ) + (1 + p1 (-1 + δ) δ) (πC - πN))) / (p δ^3 (F + πC - πN))
```

Incentive Compatibility - Collusive Strategy NSN

The profits from collusive strategy NSN can be solved as

```
FullSimplify[Solve[V == α (1/2 (πN - γS F + δ V) + 1/2 (πC + δ (p0 (πN - F) + (1 - p0) πC + δ V))) +
    (1 - α) (1/2 (πC + δ V) + 1/2 (πC + δ (πC + δ V))), V]]
```

$$\left\{\left\{V \to \frac{F\,\alpha\,(\gamma S + p0\,\delta) - 2\,\pi C - \delta\,\pi C + \alpha\,(1 + p0\,\delta)\,(\pi C - \pi N)}{-2 + \delta + \delta^2}\right\}\right\}$$

```
VNSN[F_, p0_, α_, δ_, γS_, πC_, πN_] := (F α (γS + p0 δ) - 2 πC - δ πC + α (1 + p0 δ) (πC - πN)) / (-2 + δ + δ^2)
```

Deviations at the Collusion Stage

The deviation profits at the collusion stage are the same as those considered for collusive strategies LXX and NNN above. Hence the threshold value of α at the collusion stage may be solved as

```
FullSimplify[Solve[VNSN[F, p0, α, δ, γS, πC, πN] == VD[δ, πD, πN], α]]
```

$$\left\{\left\{\alpha \to \frac{(2 + \delta)\,(\pi C + (-1 + \delta)\,\pi D - \delta\,\pi N)}{F\,(\gamma S + p0\,\delta) + (1 + p0\,\delta)\,(\pi C - \pi N)}\right\}\right\}$$

We store this as the α_{NSN} constraint.

```
αNSN[F_, p0_, δ_, γS_, πC_, πD_, πN_] := ((2 + δ) (πC + (-1 + δ) πD - δ πN)) / (F (γS + p0 δ) + (1 + p0 δ) (πC - πN))
```

Deviations at the Leniency Stage

The profits from deviating and applying for leniency as the first applicant are

```
V1SN[F_, δ_, γ1_, πN_] := πN / (1 - δ) - γ1 F
```

The continuation profits from sticking to NSN instead of applying for leniency are

```
VnSN[F_, p0_, V_, δ_, γS_, πC_, πN_] :=
  1/2 (πN - γS F + δ V) + 1/2 (πC + δ (p0 (πN - F) + (1 - p0) πC + δ V))
```

Hence the threshold in terms of α may be calculated as

```
FullSimplify[
  Solve[VnSN[F, p0, VNSN[F, p0, α, δ, γS, πC, πN], δ, γS, πC, πN] == V1SN[F, δ, γ1, πN], α]]
```

$$\left\{\left\{\alpha \to ((2 + \delta)\,(F\,(-1 + \delta)\,(-2\,\gamma 1 + \gamma S + p0\,\delta) + (1 + \delta + p0\,(-1 + \delta)\,\delta)\,(\pi C - \pi N))) / (\delta\,(1 + \delta)\,(F\,(\gamma S + p0\,\delta) + (1 + p0\,\delta)\,(\pi C - \pi N)))\right\}\right\}$$

We store this as the α_{lSN} constraint.

```
αlSN[F_, p0_, δ_, γ1_, γS_, πC_, πN_] :=
  ((2 + δ) (F (-1 + δ) (-2 γ1 + γS + p0 δ) + (1 + δ + p0 (-1 + δ) δ) (πC - πN))) /
   (δ (1 + δ) (F (γS + p0 δ) + (1 + p0 δ) (πC - πN)))
```

Deviations at the Settlement Stage - Settling a Weak Case

The value of this deviation is

```
VNSs[F_, δ_, γS_, πN_] := πN/(1 - δ) - γS F
```

The continuation profits of sticking to NSN instead of settling the weak case are

```
VNSn[F_, p0_, V_, δ_, πC_, πN_] := πC + δ (p0 (πN - F) + (1 - p0) πC + δ V)
```

Hence the critical value of α may be solved as

```
FullSimplify[
 Solve[VNSn[F, p0, VNSN[F, p0, α, δ, γS, πC, πN], δ, πC, πN] == VNSs[F, δ, γS, πN], α]]
{{α → ((2 + δ) (F (-1 + δ) (-γS + p0 δ) + (1 + p0 (-1 + δ) δ) (πC - πN))) /
     (δ^2 (F (γS + p0 δ) + (1 + p0 δ) (πC - πN)))}}
```

We store this as the α_{NSs} constraint.

```
αNSs[F_, p0_, δ_, γS_, πC_, πN_] :=
  ((2 + δ) (F (-1 + δ) (-γS + p0 δ) + (1 + p0 (-1 + δ) δ) (πC - πN))) /
   (δ^2 (F (γS + p0 δ) + (1 + p0 δ) (πC - πN)))
```

Profitability Thresholds

i) Comparing V_{NNN} and V_{LXX}

We can solve for the threshold value of p as follows

```
FullSimplify[Solve[VNNN[F, p, α, δ, πC, πN] == VLXX[F, α, δ, γL, πC, πN], {p}]]
{{p → ((1 + δ) (F γL + πC - πN))/(δ (F + πC - πN))}}
```

Store this threshold as $p^{\text{NNN}}_{\text{LXX}}$.

```
pNNNLXX[F_, δ_, γL_, πC_, πN_] := ((1 + δ) (F γL + πC - πN))/(δ (F + πC - πN))
```

ii) Comparing V_{NNN} and V_{NSN}

We can solve for the threshold value of p as follows

```
FullSimplify[Solve[VNNN[F, p, α, δ, πC, πN] == VNSN[F, P0[d, p], α, δ, γS, πC, πN], {p}]]
{{p → ((1 + δ) (F (γS - d δ) + πC - πN + d δ (-πC + πN)))/(δ (F + πC - πN))}}
```

Store this threshold as $p^{\text{NNN}}_{\text{NSN}}$.

```
pNNNNSN[F_, δ_, d_, γS_, πC_, πN_] := ((1 + δ) (F (γS - d δ) + πC - πN + d δ (-πC + πN)))/(δ (F + πC - πN))
```

iii) Comparing V_{NSN} and V_{LXX}

We can solve for the threshold value of *p* as follows

```
FullSimplify[Solve[VNSN[F, P0[d, p], α, δ, γS, πC, πN] == VLXX[F, α, δ, γL, πC, πN], {p}]]
```

$$\left\{\left\{p \to \frac{F\left(-\gamma S + d\,\delta + \gamma L\,(2+\delta)\right) + (1+\delta+d\,\delta)\,(\pi C - \pi N)}{\delta\,(F + \pi C - \pi N)}\right\}\right\}$$

Store this threshold as p^{NSN}_{LXX}.

$$\texttt{pNSNLXX[F\_, δ\_, d\_, γL\_, γS\_, πC\_, πN\_] :=} \frac{F\left(-\gamma S + d\,\delta + \gamma L\,(2+\delta)\right) + (1+\delta+d\,\delta)\,(\pi C - \pi N)}{\delta\,(F + \pi C - \pi N)}$$

Deriving Figures 1 and 2

```
Clear[F, d, γ1, γL, γS, πC, πD, πN, δ]
```

Consider the following numerical example.

```
{F = 4, d = 0.15, γ1 = 0.1, γL = 0.1, γS = 0.13, πC = 1, πD = 3, πN = 0, δ = 0.9}
{4, 0.15, 0.1, 0.1, 0.13, 1, 3, 0, 0.9}
```

Note that, at these parameter values, we fall within case (ii) of Proposition 4, since δ_S is equal to

$$N\left[\frac{F\,(-\gamma L + \gamma S)}{d\,(F + \pi C - \pi N)}\right]$$

```
0.16
```

which lies below the specified value of δ=0.9. Hence, all three of the collusive strategies may be value maximizing.

```
RegionPlot[{α < αLXX[δ, γL, πC, πD, πN, F],
  α < Min[αNSN[F, P0[d, p], δ, γS, πC, πD, πN], α1SN[F, P0[d, p], δ, γ1, γS, πC, πN],
    αNSs[F, P0[d, p], δ, γS, πC, πN]], α < Min[αNNN[F, p, δ, πC, πD, πN],
    α1NN[F, p, δ, γ1, πC, πN], αNsN[F, p, P1[d, p], δ, γS, πC, πN]]}, {p, d, 1 - d},
 {α, 0, 1}, FrameLabel → {"p", "α"}, PlotLegends → {"L", "NSN", "NNN"},
 PlotRange → {0, 1}, Epilog → {Black,
   Line[{{pNSNLXX[F, δ, d, γL, γS, πC, πN], 0}, {pNSNLXX[F, δ, d, γL, γS, πC, πN], 1}}],
   Line[{{pNNNLXX[F, δ, γL, πC, πN], 0}, {pNNNLXX[F, δ, γL, πC, πN], 1}}],
   Line[{{pNNNNSN[F, δ, d, γS, πC, πN], 0}, {pNNNNSN[F, δ, d, γS, πC, πN], 1}}]}]
```



The vertical lines correspond to

```
pNNNLXX[F, δ, γL, πC, πN]
```

```
0.591111
```

```
pNNNNSN[F, δ, d, γS, πC, πN]
```

```
0.356778
```

```
pNSNLXX[F, δ, d, γL, γS, πC, πN]
```

```
0.714444
```

We can show that the binding constraint on collusive strategies NSN and NNN arises at the collusion stage for all *p*.

```
Plot[{αNSN[F, P0[d, p], δ, γS, πC, πD, πN],
  α1SN[F, P0[d, p], δ, γ1, γS, πC, πN], αNSs[F, P0[d, p], δ, γS, πC, πN], 1},
 {p, d, 1 - d}, PlotRange → {{0, 1}, {0, 3}}, PlotLabels → {"αNSN", "α1SN", "αNSs", 1}]
```




```
Plot[{αNNN[F, p, δ, πC, πD, πN], α1NN[F, p, δ, γ1, πC, πN],
  αNsN[F, p, P1[d, p], δ, γS, πC, πN], 1}, {p, d, 1 - d},
 PlotRange → {{0, 1}, {0.2, 1.1}}, PlotLabels → {"αNNN", "α1NN", "αNsN", 1}]
```



Hence we can plot Figure 1 in formatted form as follows

```
F1 = Show[ContourPlot[{α - αLXX[δ, γL, πC, πD, πN, F]}, {p, d, 1 - d},
    {α, 0, 1}, FrameLabel → {p, α}, Contours → {0}, ContourShading → None,
    RotateLabel → False, LabelStyle → {FontSize → 17, FontFamily → "Times", Black},
    ContourStyle → {{Thick, Black, AbsoluteDashing[6]}}, PlotRangePadding → .0,
    FrameTicks → {{{{0, "0.0"}, 0.2, 0.4, 0.6, 0.8, {1, "1.0"}}, None},
      {{0.1, 0.3, 0.5, 0.7, 0.9}, None}}, PlotRange → {{0, 1}, {0, 1}}],
   ContourPlot[{p == pNSNLXX[F, δ, d, γL, γS, πC, πN]}, {p, d, 1 - d},
    {α, 0, 0.88}, ContourStyle → {Gray, Thin}],
   ContourPlot[{p == pNNNLXX[F, δ, γL, πC, πN]}, {p, d, 1 - d},
    {α, 0, 1}, ContourStyle → {Gray, Thin}],
   ContourPlot[{p == pNNNNSN[F, δ, d, γS, πC, πN]}, {p, d, 1 - d},
    {α, 0, 1}, ContourStyle → {Gray, Thin}],
   ContourPlot[{α - αNSN[F, P0[d, p], δ, γS, πC, πD, πN]}, {p, d, 1 - d}, {α, 0, 1},
    Contours → {0}, ContourShading → None, ContourStyle → {{Black, Thick}}],
   RegionPlot[{p < d, p > 1 - d}, {p, 0, 1}, {α, 0, 1}, PlotRange → {{0, 1}, {0, 1}},
    PlotStyle → GrayLevel[0.95], BoundaryStyle → GrayLevel[0.95]],
   ContourPlot[{α - αNNN[F, p, δ, πC, πD, πN]}, {p, d, 1 - d}, {α, 0, 1}, Contours → {0},
    ContourShading → None, ContourStyle → {{Dotted, Thick}}, ContourLabels → Function[
      {x, y}, {Text[Style["p^NSN_LXX", FontSize → 17, FontFamily → "Times"], {0.79, 0.83}],
       Text[Style["p^NNN_LXX", FontSize → 17, FontFamily → "Times"], {0.668, 0.95}],
       Text[Style["p^NNN_NSN", FontSize → 17, FontFamily → "Times"], {0.432, 0.95}], Text[
        Style[Subscript["α", "NSN"], FontSize → 17, FontFamily → "Times"], {0.93, 0.43}],
       Text[Style[Subscript["α", "NNN"], FontSize → 17, FontFamily → "Times"],
        {0.93, 0.34}], Text[Style[Subscript["α", "LXX"],
         FontSize → 17, FontFamily → "Times"], {0.93, 0.5}]}]]
 ]
```



```
F2 = Show[
  RegionPlot[{p < d, p > 1 - d}, {p, 0, 1}, {α, 0, 1}, PlotRange → {{0, 1}, {0, 1}},
   PlotStyle → GrayLevel[0.95], BoundaryStyle → GrayLevel[0.95], RotateLabel → False,
   LabelStyle → {FontSize → 17, FontFamily → "Times", Black}, FrameLabel → {p, α},
   FrameTicks → {{{{0, "0.0"}, 0.2, 0.4, 0.6, 0.8, {1, "1.0"}}, None},
     {{0.1, 0.3, 0.5, 0.7, 0.9}, None}}],
  RegionPlot[{α < αNNN[F, p, δ, πC, πD, πN] && p < pNNNNSN[F, δ, d, γS, πC, πN]},
   {p, d, 1 - d}, {α, 0, 1}, PlotRange → {{0.45, 0.9}, {0, 0.8}},
   PlotStyle → GrayLevel[1], BoundaryStyle → {Black, Dotted}, PlotPoints → 35],
  RegionPlot[{α < αLXX[δ, γL, πC, πD, πN, F] && p > pNSNLXX[F, δ, d, γL, γS, πC, πN]},
   {p, d, 1 - d}, {α, 0, 1}, PlotRange → {{0.45, 0.9}, {0, 0.8}}, PlotStyle → GrayLevel[1],
   BoundaryStyle → {Black, AbsoluteDashing[6]}, PlotPoints → 35],
  RegionPlot[α < αNSN[F, P0[d, p], δ, γS, πC, πD, πN] &&
    pNNNNSN[F, δ, d, γS, πC, πN] ≤ p ≤ pNSNLXX[F, δ, d, γL, γS, πC, πN],
   {p, d, 1 - d}, {α, 0, 1}, PlotRange → {{0, 1}, {0, 1}},
   PlotStyle → GrayLevel[1], BoundaryStyle → {Black}, PlotPoints → 35],
  ContourPlot[{α - αNNN[F, p, δ, πC, πD, πN]}, {p, d, 1 - d}, {α, 0, 1},
   FrameLabel → {p, α}, Contours → {0}, ContourShading → None,
   RotateLabel → False, LabelStyle → {FontSize → 17, FontFamily → "Times", Black},
   ContourStyle → {{Opacity[0]}}, ContourLabels → Function[{x, y},
     {Text[Style["NNN", FontSize → 17, FontFamily → "Times"], {0.25, 0.25}],
      Text[Style["NSN", FontSize → 17, FontFamily → "Times"], {0.53, 0.25}],
      Text[Style["NC", FontSize → 17, FontFamily → "Times"], {0.78, 0.95}],
      Text[Style["LXX", FontSize → 16, FontFamily → "Times"], {0.785, 0.25}]}],
   PlotRangePadding → .0, FrameTicks → {{{{0, "0.0"}, 0.2, 0.4, 0.6, 0.8, {1, "1.0"}},
      None}, {{0.1, 0.3, 0.5, 0.7, 0.9}, None}},
   PlotRange → {{0.45, 0.9}, {0, 0.8}}, PlotRangePadding → .0]
 ]
```



We contrast this outcome with that arising in the absence of settlements. To that end set

```
γS = 99 999 999 999 999
```

```
99 999 999 999 999
```

```
RegionPlot[{α < αLXX[δ, γL, πC, πD, πN, F],
  α < Min[αNSN[F, P0[d, p], δ, γS, πC, πD, πN], α1SN[F, P0[d, p], δ, γ1, γS, πC, πN],
    αNSs[F, P0[d, p], δ, γS, πC, πN]], α < Min[αNNN[F, p, δ, πC, πD, πN],
    α1NN[F, p, δ, γ1, πC, πN], αNsN[F, p, P1[d, p], δ, γS, πC, πN]]}, {p, d, 1 - d},
 {α, 0, 1}, FrameLabel → {"p", "α"}, PlotLegends → {"L", "NSN", "NNN"},
 PlotRange → {0, 1}, Epilog → {Black,
   Line[{{pNSNLXX[F, δ, d, γL, γS, πC, πN], 0}, {pNSNLXX[F, δ, d, γL, γS, πC, πN], 1}}],
   Line[{{pNNNLXX[F, δ, γL, πC, πN], 0}, {pNNNLXX[F, δ, γL, πC, πN], 1}}],
   Line[{{pNNNNSN[F, δ, d, γS, πC, πN], 0}, {pNNNNSN[F, δ, d, γS, πC, πN], 1}}]}]
```



Of course, the binding constraint on collusive strategy NNN still arises at the collusion stage.

```
Plot[{αNNN[F, p, δ, πC, πD, πN], α1NN[F, p, δ, γ1, πC, πN],
  αNsN[F, p, P1[d, p], δ, γS, πC, πN], 1}, {p, d, 1 - d},
 PlotRange → {{0, 1}, {0.2, 1.1}}, PlotLabels → {"αNNN", "α1NN", 1}]
```



The vertical line corresponds to

```
pNNNLXX[F, δ, γL, πC, πN]
0.591111
```

The remaining two thresholds no longer play any role

```
pNNNNSN[F, δ, d, γS, πC, πN]
```

1.68889×10^{14}

```
pNSNLXX[F, δ, d, γL, γS, πC, πN]
```

-8.88889×10^{13}

In formatted form

```
Show[
 ContourPlot[{α - αLXX[δ, γL, πC, πD, πN, F]}, {p, d, 1 - d},
  {α, 0, 1}, FrameLabel → {p, α}, Contours → {0}, ContourShading → None,
  RotateLabel → False, LabelStyle → {FontSize → 17, FontFamily → "Times", Black},
  ContourStyle → {{Thick, Black, AbsoluteDashing[6]}}, PlotRangePadding → .0,
  FrameTicks → {{{{0, "0.0"}, 0.2, 0.4, 0.6, 0.8, {1, "1.0"}}, None},
    {{0.1, 0.3, 0.5, 0.7, 0.9}, None}}, PlotRange → {{0, 1}, {0, 1}}],
 ContourPlot[{p == pNSNLXX[F, δ, d, γL, γS, πC, πN]}, {p, d, 1 - d},
  {α, 0, 0.88}, ContourStyle → {Gray, Thin}],
 ContourPlot[{p == pNNNLXX[F, δ, γL, πC, πN]}, {p, d, 1 - d},
  {α, 0, 1}, ContourStyle → {Gray, Thin}],
 ContourPlot[{p == pNNNNSN[F, δ, d, γS, πC, πN]}, {p, d, 1 - d},
  {α, 0, 1}, ContourStyle → {Gray, Thin}],
 RegionPlot[{p < d, p > 1 - d}, {p, 0, 1}, {α, 0, 1}, PlotRange → {{0, 1}, {0, 1}},
  PlotStyle → GrayLevel[0.95], BoundaryStyle → GrayLevel[0.95]],
 ContourPlot[{α - αNNN[F, p, δ, πC, πD, πN]}, {p, d, 1 - d},
  {α, 0, 1}, Contours → {0}, ContourShading → None,
  ContourStyle → {{Dotted, Thick}}, ContourLabels → Function[{x, y},
    {Text[Style["p^NNN_LXX", FontSize → 17, FontFamily → "Times"], {0.665, 0.95}], Text[
      Style[Subscript["α", "NNN"], FontSize → 17, FontFamily → "Times"], {0.93, 0.34}],
     Text[Style[Subscript["α", "LXX"], FontSize → 17, FontFamily → "Times"],
      {0.93, 0.5}]}]]
]
```



```
Show[
 RegionPlot[{p < d, p > 1 - d}, {p, 0, 1}, {α, 0, 1}, PlotRange → {{0, 1}, {0, 1}},
  PlotStyle → GrayLevel[0.95], BoundaryStyle → GrayLevel[0.95], RotateLabel → False,
  LabelStyle → {FontSize → 17, FontFamily → "Times", Black}, FrameLabel → {p, α},
  FrameTicks → {{{{0, "0.0"}, 0.2, 0.4, 0.6, 0.8, {1, "1.0"}}, None},
    {{0.1, 0.3, 0.5, 0.7, 0.9}, None}}],
 RegionPlot[{α < αNNN[F, p, δ, πC, πD, πN] && p < pNNNLXX[F, δ, γL, πC, πN]},
  {p, d, 1 - d}, {α, 0, 1}, PlotRange → {{0.45, 0.9}, {0, 0.8}},
  PlotStyle → GrayLevel[1], BoundaryStyle → {Black, Dotted}, PlotPoints → 35],
 RegionPlot[{α < αLXX[δ, γL, πC, πD, πN, F] && p > pNNNLXX[F, δ, γL, πC, πN]},
  {p, d, 1 - d}, {α, 0, 1}, PlotRange → {{0.45, 0.9}, {0, 0.8}}, PlotStyle → GrayLevel[1],
  BoundaryStyle → {Black, AbsoluteDashing[6]}, PlotPoints → 35],
 ContourPlot[{α - αNNN[F, p, δ, πC, πD, πN]}, {p, d, 1 - d}, {α, 0, 1},
  FrameLabel → {p, α}, Contours → {0}, ContourShading → None,
  RotateLabel → False, LabelStyle → {FontSize → 17, FontFamily → "Times", Black},
  ContourStyle → {{Opacity[0]}}, ContourLabels → Function[{x, y},
    {Text[Style["NNN", FontSize → 17, FontFamily → "Times"], {0.275, 0.25}],
     Text[Style["LXX", FontSize → 16, FontFamily → "Times"], {0.775, 0.25}],
     Text[Style[1, FontSize → 17, FontFamily → "Times"], {0.65, 0.25}],
     Text[Style["NC", FontSize → 17, FontFamily → "Times"], {0.78, 0.95}],
     Text[Style[2, FontSize → 17, FontFamily → "Times"], {0.49, 0.25}],
     Text[Style[3, FontSize → 17, FontFamily → "Times"], {0.6, 0.527}]}],
  PlotRangePadding → .0, FrameTicks → {{{{0, "0.0"}, 0.2, 0.4, 0.6, 0.8, {1, "1.0"}},
     None}, {{0.1, 0.3, 0.5, 0.7, 0.9}, None}},
  PlotRange → {{0.45, 0.9}, {0, 0.8}}, PlotRangePadding → .0],
 ContourPlot[{α - αNSN[F, P0[d, p], δ, 0.15, πC, πD, πN]},
  {p, pNNNNSN[F, δ, d, 0.15, πC, πN], pNSNLXX[F, δ, d, γL, 0.15, πC, πN]},
  {α, 0, 1}, Contours → {0}, ContourShading → None,
  ContourStyle → {{Black, Thin}}],
 ContourPlot[{p == pNNNNSN[F, δ, d, 0.15, πC, πN]}, {p, d, 1 - d},
  {α, 0, αNNN[F, pNNNNSN[F, δ, d, 0.15, πC, πN], δ, πC, πD, πN]},
  ContourStyle → {Black, Thin}],
 ContourPlot[{p == pNSNLXX[F, δ, d, γL, 0.15, πC, πN]}, {p, d, 1 - d},
  {α, 0, αLXX[δ, γL, πC, πD, πN, F]}, ContourStyle → {Black, Thin}]
]
```




```
Clear[F, d, γ1, γL, γS, πC, πD, πN, δ]
```